

A Review of Matrix Algebra for Power and Energy Applications

Alejandro Garcés-Ruiz

Universidad Tecnológica de Pereira
Carrera 27 nro. 10-02 Los Álamos-Pereria-Risaralda-Colombia
www.utp.edu.co

Abstract

This report presents a brief review of matrix algebra and its implementation in Julia for power and energy applications. First, we present basic examples of data visualization, followed by conventional operations with matrices and vectors. Then, we study quadratic forms and norms, two main concepts required in the convergence study of the power flow in power and energy applications. After that, we give good practices to create a neat code in Julia. There is an extensive set of examples available on the Internet related to these basic aspects, so we avoid repeating what is well documented. Hence, we show only basic examples to create our first scripts.

Contents

1	An introduction to Julia	1
2	Vectors, matrices and hyper-arrays	2
3	Linear transformations, eigenvalues, and eigenvectors	5
4	Quadratic forms, semidefinite matrices and norms	7
5	Linear systems and sparse matrices	10
6	Data frames	12
7	Good practices	13
8	Further readings	13

1 An introduction to Julia

Julia is a high-level, high-performance programming language designed for technical computing. It offers several advantages, which have contributed to its growing popularity in power systems applications. The syntax is easy to learn and read, since it resembles other high-level languages like Matlab and Python, two of the most common languages for programming computational methods in power systems and power distribution networks. However, Julia outperforms these languages, approaching the performance of low-level languages like C and Fortran.

Like Python, it is an open-source project with an active community of developers and researchers who create packages, and provide support through forums, web pages, and mailing lists. There are different environments for developing Julia code such as Visual Studio Code, Juno and Jupyter Notebooks. We do not catch ourselves with any of these environments, since Julia is a constantly evolving ecosystem. We recommend checking the latest information on Julia's official website and the community resources for the most up-to-date details. Let's introduce our first examples:

•• Example 1.

Julia has a built-in package manager that ensures seamless integration with external libraries and tools. A new package is added using the package manager as follows:

```
using Pkg
Pkg.add("Plots")
```

This code installs the package `Plots` which is an extensive library for data visualization. The next example shows its usage.

•• Example 2.

`Plots` is a library that resembles Matlab functionalities. Once the package is installed, we can plot a simple function as given below:

```
using Plots
x = LinRange(0,2*pi,100)
y = sin.(x)
plot(x,y)
```

The first line calls the package `Plots` that allows data visualization, while the second line generates an array with 100 points between 0 and 2π . Julia allows special characters into the code, so we can replace the word `pi` by the Greek letter π , obtaining the same result. The third line applies the function `sin` to each point x . Finally, the fourth line plots x vs y . The reader is invited to implement this code to see the result. Notice the function `sin` is evaluated in each point of the array x . Hence, it requires a dot to indicate it is a pointwise operation.

•• Example 3.

Julia resembles both Python and Matlab; thus, it is easy to translate a code from any of these languages to Julia. Let us see two ways to define a for loop and print the iteration:

```
for k = 1:10
    println("iteration:",k)
end

for k in 1:10
    println("iteration:",k)
end
```

Both codes are functionally equivalent, and the choice between them depends on the specific application and personal preferences. Readers familiar with Matlab may prefer the first code, whereas those accustomed to Python may favor the second.

•• Example 4.

We can also define a loop inside an array in order to define its elements, namely:

```
using Plots
v = [cos(t)+sin(t)*1im for t in 0:0.1:2*pi]
plot(v)
```

Notice the imaginary number is given by `1im`. In this case, we have created a complex vector $v = \cos(t) + i \sin(t)$. Then, we plot the real and imaginary parts, obtaining a unitary circle.

2 Vectors, matrices and hyper-arrays

Some physical magnitudes, such as velocity or acceleration, are characterized by having not only magnitude but also direction. These magnitudes are called vectors and are usually defined in the three-dimensional space; however, we can generalize the concept to arrays in \mathbb{R}^n , \mathbb{C}^n . In the same way, we can build two-dimensional




Figure 1: Example of a vector, a matrix, and a three-dimensional hyper-array

arrays, that is, matrices and even three-dimensional arrays (tensors or hyper-arrays) that resemble Rubik's cubes, as shown in Figure 1.

These arrays are easily represented in Julia, as presented in the following example:

•• **Example 5.**

This example creates random arrays of different sizes. Entries of these arrays are obtained using square brackets, for instance: $A[1]$, $B[4, 1]$, $C[3, 2, 1]$; other outputs are given below:

```
A = rand(5)
B = rand(5, 2)
C = rand(3, 3, 3)
println(" Vector A =", A)
println(" Matrix B =", B)
println(" Hyper-array C =", C)
println(" First column of B =", B[:, 1])
println(" Second row of B =", B[2, :])
println(" First matrix of C =", C[:, :, 1])
```

Most of the problems in power systems are represented in high dimensional spaces. Therefore, we require storing and operate with vectors, matrices and hyper-arrays. Basic operations such as adding and multiplying by a scalar are intuitively defined for both, real and complex domains.

•• **Example 6.**

This example shows basic operations with complex arrays:

```
A = [3+2im 8+3im]
B = [1-4im 7+5im]
C = A+B
println(C)
println(C')
println(transpose(C))
```

Notice the difference between C' and $\text{transpose}(C)$. The former returns the transpose conjugate, whereas the latter returns the transpose without conjugate.

We can define other operators such the trace, determinant, and transpose for matrix variables. Let us revisit briefly these concepts:

The trace of a square matrix A is denoted by $\text{tr}(A)$. It is a function $\text{tr} : \mathbb{R}^{n \times n} \rightarrow \mathbb{R}$ that takes the sum of the elements in the diagonal, namely:

$$\text{tr}(A) = \sum_{k=1}^n a_{kk} \quad (1)$$

The following properties hold for any pair of square matrices A, B and, a scalar α :

$$\text{tr}(A + B) = \text{tr}(A) + \text{tr}(B) \quad (2)$$

$$\text{tr}(\alpha A) = \alpha \text{tr}(A) \quad (3)$$

$$\text{tr}(A^T) = \text{tr}(A) \quad (4)$$

$$(A^T)^T = A \quad (5)$$

$$(A + B)^T = A^T + B^T \quad (6)$$

$$(AB)^T = B^T A^T \quad (7)$$

The determinant is another matrix function $\det : \mathbb{R}^{n \times n} \rightarrow \mathbb{R}$, that takes the n column (or row) vectors that form a matrix, and returns a real number $\det(A) = \det([a_1, a_2, \dots, a_n])$. There are several interpretations of the determinant. Perhaps the most useful, and visually meaningful of these interpretations, is the volume of the parallelogram generated by the vectors that form the matrix (see Figure 2).




Figure 2: Example of a parallelogram (\mathbb{R}^2) or a parallelepiped (\mathbb{R}^3) formed by a set of linearly independent vectors.

The area of the parallelogram \mathcal{P} formed by the vectors v_1 and v_2 can be calculated by organizing these vectors into a matrix A as given below:

$$\text{Area}(\mathcal{P}) = |\det(A)| \quad (8)$$

This concept can be extended to the volume of \mathcal{P} for the three-dimensional case and the hyper-volume for higher dimensions.

Interpreting the determinant as a volume helps to remember and better understand some of its properties. For example, $\det(A) = 0$ if a_1 and a_2 are linearly dependent because in that case, both vectors are parallel (or antiparallel) and do not generate any area. The determinant of the identity matrix is $\det(I) = 1$ because it represents a square generated by two vectors of size one. Other properties are given below:

$$\det(A^T) = \det(A) \quad (9)$$

$$\det(A^{-1}) = 1/\det(A) \quad (10)$$

$$\det(AB) = \det(A)\det(B) \quad (11)$$

$$\det(P^{-1}AP) = \det(A) \quad (12)$$

$$\det(A) \neq 0, \text{ if } A \text{ is non-singular} \quad (13)$$

●● Example 7.

The package `LinearAlgebra` allows operating with vectors and matrices. The following code defines a 2×2 matrix of zeros and calculates its trace, determinant, and rank:

```
using LinearAlgebra
M = zeros(2,2)
d = det(M)
t = tr(M)
r = rank(M)
printstyled("determinant of M \ t = ",
d, "\n", color = :red)
printstyled("trace of M \ t = ",t, "\n",
color = :green)
```

```
printstyled("rank of M \t = ",r, "\n",
color = :blue)
```

Additional features of Julia were showcased in this script. Comments are defined with the number symbol (hash-tag); character `\t` inserts a tap, while `\n` creates a new line; `printstyled` allows printing the results with different colors and format.

Besides the conventional matrix multiplication, we can also define the Hadamard and Kronecker product. Given two matrices A and B , the Hadamard product, denoted by $A \circ B$, is the obtained by an element-wise multiplication; for this, the size of both matrices must be the same. The Kronecker product returns an augmented matrix, denoted by $A \otimes B$, where each element of A is multiplied by the entire matrix B .

●● **Example 8.**

Consider two matrices A and B defined as follows:

$$A = \begin{bmatrix} 1 & 2 \\ 8 & 5 \end{bmatrix} \quad (14)$$

$$B = \begin{bmatrix} 3 & 4 \\ 7 & 6 \end{bmatrix} \quad (15)$$

The following code returns the conventional product, the Hadamard product and, the Kronecker product:

```
using LinearAlgebra
A = [1 2; 8 5]
B = [3 4; 7 6]
println("Conventional product\t =", A*B)
println("Hadamard product\t =", A.*B)
println("Kronecker product\t =", kron(A,B))
```

For the sake of better understanding the concept, we present each of these operations step by step:

$$AB = \begin{bmatrix} (1)(3) + (2)(7) & (1)(4) + (2)(6) \\ (8)(3) + (5)(7) & (8)(4) + (5)(6) \end{bmatrix} = \begin{bmatrix} 17 & 16 \\ 59 & 62 \end{bmatrix} \quad (16)$$

$$A \circ B = \begin{bmatrix} (1)(3) & (2)(4) \\ (8)(7) & (5)(6) \end{bmatrix} = \begin{bmatrix} 3 & 8 \\ 56 & 30 \end{bmatrix} \quad (17)$$

$$A \otimes B = \begin{bmatrix} (1) \begin{bmatrix} 3 & 4 \\ 7 & 6 \end{bmatrix} & (2) \begin{bmatrix} 3 & 4 \\ 7 & 6 \end{bmatrix} \\ (8) \begin{bmatrix} 3 & 4 \\ 7 & 6 \end{bmatrix} & (5) \begin{bmatrix} 3 & 4 \\ 7 & 6 \end{bmatrix} \end{bmatrix} = \begin{bmatrix} 3 & 4 & 6 & 8 \\ 7 & 6 & 14 & 12 \\ 24 & 32 & 15 & 20 \\ 56 & 48 & 35 & 30 \end{bmatrix} \quad (18)$$

Very complicated operations in power systems can be represented and analyzed by a Hadamard or a Kronecker product.

It is important to identify the properties of each of these products. For instance, conventional matrix multiplication requires that the number of columns of the first matrix is the same as the number of rows of the second matrix. This multiplication is usually non-commutative ($AB \neq BA$). The Hadamard product requires two matrices of the same size. This product is commutative, i.e., $A \circ B = B \circ A$. The Kronecker product allows matrices of any size. Its result is non-commutative.

3 Linear transformations, eigenvalues, and eigenvectors

A linear transformation is a mathematical operation that takes input points and transforms them, creating another space where, perhaps, it is easy to make calculations. This transformation involves stretching, squeezing, rotating, reflecting, or shearing the points. However, it must follow certain rules such as preserving the origin,

and maintaining addition and scalar multiplication. Linear transformations appear in several power systems' computation problems. These include the Park and Clarke transformation as well as the symmetrical components.

Matrix transformation are usually represented by a transformation matrix A . Thus, each vector x from the original space can be transformed into a new vector $z = Ax$. Geometric properties of the transformed space are given by A . There exist some vectors v which maintain their direction in the original and the transformed space. That is to say, $Av = \lambda v$. These are the eigenvectors of A and the proportional constant λ are the eigenvalues. Every square matrix has a set of unique eigenvalues. However, eigenvectors are not unique, since every multiple is also an eigenvector. Besides, the following two properties hold:

$$\det(A) = \prod_{i=1}^n \lambda_i \quad (19)$$

$$\text{tr}(A) = \sum_{i=1}^n \lambda_i \quad (20)$$

Both, eigenvectors and eigenvalues play a key role in stability problems for power systems applications. Let us consider an example where both concepts are presented in the context of a linear transformation.

•• Example 9.

Consider a set of points representing a unit circle in the plane:

$$\Omega = \{x \in \mathbb{R}^2 : x_1^2 + x_2^2 = 1\} \quad (21)$$

This set can be created parametrically as given below:

```
using LinearAlgebra
using Plots
Omega = zeros(2,100)
for k = 1:100
    Omega[1,k] = cos(2*pi*k/100)
    Omega[2,k] = sin(2*pi*k/100)
end
p1 = plot(Omega[1,:],Omega[2,:],aspect_ratio=:equal)
```

Consider now a linear transformation represented by a 2×2 matrix A , and a new space Γ as follows:

$$\Gamma = \{z \in \mathbb{R}^2 : z = Ax, \forall x \in \Omega\} \quad (22)$$

This set can also be defined and plotted in Julia:

```
A = [1 0.3;0.5 1]
Gamma = zeros(2,100)
for k = 1:100
    Gamma[:,k] = A*Omega[:,k]
end
p2 = plot(Gamma[1,:],Gamma[2,:],aspect_ratio=:equal)
```

Finally, we can calculate the eigenvalues and eigenvectors of the matrix transformation. The eigenvectors are the directions that maintain unchanged after the transformation.

```
(L,V) = eigen(A)
p2=plot!([0,V[1,1]],[0,V[1,2]],arrow=1)
p2=plot!([0,V[2,1]],[0,V[2,2]],arrow=1)
plot(p1,p2,layout=(1,2))
```

Figure 3 shows the original and the transformed space. The former contains a circle that is transformed into an ellipse. The eigenvalues are the major axis of this ellipse.




Figure 3: Example of a linear transformation. The eigenvectors v represent the direction which remains after the transformation.

4 Quadratic forms, semidefinite matrices and norms

A quadratic form is a function from \mathbb{R}^n to \mathbb{R} defined by the following matrix equation:

$$q(x) = x^T M x \quad (23)$$

where M is a square matrix. This type of function is a polynomial where the maximum power is two. For example, the following quadratic equation:

$$q(x_1, x_2) = x_1^2 + 3x_1x_2 - x_2^2 \quad (24)$$

can be represented by a function as (23) with,

$$M = \begin{bmatrix} 1 & 3 \\ 0 & -1 \end{bmatrix} \quad (25)$$

A quadratic form q can be represented by different matrices M . For example, we can replace M by a symmetric matrix, $(M^T + M)/2$ obtaining the same polynomial, namely:

$$q(x_1, x_2) = \begin{bmatrix} x_1 \\ x_2 \end{bmatrix}^T \begin{bmatrix} 1 & 3/2 \\ 3/2 & -1 \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \end{bmatrix} \quad (26)$$

Quadratic functions appear in many applications in power systems, among which stand out the power loss calculation using the nodal admittance matrix.

•• Example 10.

The following script creates a quadratic form using a random matrix $M \in \mathbb{R}^{2 \times 2}$. Notice how a function is created and used in Julia.

```
function q(x)
    M = [3.5 4.2;8.1 5.0]
    return transpose(x)*M*x
end
x0 = [1,1]
y0 = q(x0)
println(y0)
```

The operation `transpose(x)` can be replaced by x' with the same result. However, it is important to be careful in the case of complex arrays where x' transpose and conjugate. Therefore, it is recommended to use the command `transpose` to be certain we are only transposing without conjugating.

•• Example 11.

The shape of a quadratic form may vary significantly according to the properties of its generating matrix. Figure

4 shows two different quadratic forms: the first one generated by the matrix (25) and the second one by the following matrix:

$$M = \begin{bmatrix} 0.6 & 0.2 \\ 0.2 & 0.8 \end{bmatrix} \quad (27)$$

The second plot is a paraboloid, which is a convex function generated by a positive definite matrix (a concept that will be defined below). These plots can be generated in Julia as follows:




Figure 4: Example of two different quadratic forms: left) $q(x_1, x_2) = x_1^2 + 3x_1x_2 - x_2^2$, right) $q(x_1, x_2) = 0.6x_1^2 + 0.4x_1x_2 + 0.8x_2^2$. The latter is a paraboloid defined by a quadratic form with a positive definite matrix.

```
using Plots
xs = range(-10, stop=10, length=100)
ys = range(-10, stop=10, length=100)
M1 = [0.6 0.2; 0.2 0.8]
f1(x1, x2) = [x1; x2]' * M1 * [x1; x2]
M2 = [1 3; 0 -1]
f2(x1, x2) = [x1; x2]' * M2 * [x1; x2]
plt1 = surface(xs, ys, f1)
plt2 = contourf(xs, ys, f1)
plt3 = surface(xs, ys, f2)
plt4 = contourf(xs, ys, f2)
plot(plt1, plt2, plt3, plt4)
```

We say that M is positive definite, represented by $M > 0$, if $q(x) > 0$ for any $x \in \mathbb{R}^n$ different from zero. Positive definite matrices are non-singular, meaning they have an inverse and its determinant is different from zero. On the other hand, a matrix is positive semidefinite, i.e., $M \geq 0$, if $q(x) \geq 0$, $\forall x \neq 0$. Positive definite and semidefinite matrices play a crucial role in many applications. For example, a quadratic form with a positive semidefinite matrix is convex. This property is important in optimization problems since convexity allows guaranteeing global optimum and convergence of the algorithms [Garcés, 2022]. It also helps to demonstrate convergence in the fixed-point and newton methods for the power flow analysis [?].

•• Example 12.

All eigenvalues of a positive definite matrix are positive. Hence, we can calculate the eigenvalues to determine if a matrix is positive definite. The following script calculates the eigenvalues L and eigenvectors V of a 2×2 matrix A :

```
using LinearAlgebra
A = [3 -2; -2 8]
(L,V) = eigen(A)
println(" Eigenvalues \t=",L)
println(" Eigenvectors \t=",V)
if minimum(L)>0
```



```
println("The matrix is positive definite")
end
```

This example was also an excuse to show the use of the `if` statement, which is intuitively defined in Julia.

We have the option to determine whether a matrix is positive definite by calculating its eigenvalues. However, certain matrices possess specific structural properties that guarantee positivity (or negativity) definiteness without the need to compute their eigenvalues. For instance, a matrix that is strictly diagonally dominant qualifies as positive definite. This type of matrix has the following property:

$$a_{kk} > \sum_{m \neq k} |a_{km}| \quad (28)$$

where a_{kk} represents the diagonal entries of a matrix A and, a_{km} represents the non-diagonal entries ($m \neq k$). If the inequality is not strict, that is, if we replace $>$ by \geq in (28), then we say the matrix is weakly diagonal dominant. The Y_{bus} is the archetypical example in power systems. Weakly diagonal matrices are positive definite, but this holds true only when they are also weakly chained (see [Azimzadeh and Forsyth, 2016]). If they fail to meet this additional condition, then we can only guarantee that they are positive semidefinite.

A positive definite matrix is non-singular and hence, the inverse is well-defined. Moreover, it can be represented by $A = C^T C$ where C is a triangular matrix named Cholesky factorization; C is represented as $A^{1/2}$ since $A = C^T C$ (not to confuse with a elementwise square root of A). This factorization can be used for fast solving of linear systems of the form $Ax = b$.

Like a quadratic form, a norm is another type of function that takes objects from \mathbb{R}^n and returns a real number. Intuitively, a norm is a way to measure the size of a vector. For instance, the Euclidean norm, which is the most conventional of these functions defined as follows:

$$\|x\| = \sqrt{x^T x} \quad (29)$$

However, we can define a Q-norm from any quadratic function defined by a matrix Q as long as this matrix is positive definite:

$$\|x\|_Q = \sqrt{x^T Q x}, \quad Q > 0 \quad (30)$$

Similar to the Euclidean norm, $\|x\|_Q$ is positive for any x except for the vector $x = 0$. In addition, the triangular inequality holds, that is:

$$\|x + y\| \leq \|x\| + \|y\| \quad (31)$$

Clearly, the numerical result of $\|x\|$ is different from $\|x\|_Q$; however, both functions are suitable ways to measure the size of a vector $x \in \mathbb{R}^n$. We may define a norm according to the application. Thus, Q can be regarded as a weight factor.

Similarly, we can define a matrix norm as a manner to measure the size of a matrix. We induce a matrix norm given a vector norm as given below:

$$\|A\| = \sup_{x \neq 0} \frac{\|Ax\|}{\|x\|} \quad (32)$$

For any matrix A and a vector x , we have the following property:

$$\|Ax\| \leq \|A\| \|x\| \quad (33)$$

Finally, the Frobenius norm is another way to define a size of a matrix. This norm is defined as follows:

$$\|A\|_F = \left(\sum_{i=1}^n \sum_{j=1}^n a_{ij}^2 \right)^{1/2} = \sqrt{\text{tr}(AA^T)} \quad (34)$$

Both, the induced norm given by (32) and the Frobenius norm given by (34) satisfy the following property:

$$\|AB\| \leq \|A\| \|B\| \quad (35)$$

In this case, we say that the matrix norm is sub-multiplicative.

•• Example 13.

Let us see the sub-multiplicative norm in practice. The following script generates two random matrices A and B ; then, the norm of each one is calculated and compared:

```
using LinearAlgebra
A = rand(2,2)
B = rand(2,2)
n_a = norm(A)
n_b = norm(B)
n_ab = norm(A*B)
println("norm(AB)=", n_ab,
        " < norm(A)norm(B)=", n_a*n_b)
```

Frobenius norm is the default matrix-norm in Julia. Induced 2-norm can be obtained by the command `opnorm(A, 2)` or simply `opnorm(A)`.

5 Linear systems and sparse matrices

Most of the power flows methods in for active distribution networks require solving a system of linear equations as follows:

$$Ax = b \tag{36}$$

where A is a square matrix and b is a column vector. Like most of the modern programming languages, Julia has methods for efficiently solving this type of problem. The most intuitive although less efficient way to solve (36) is, perhaps, by calculating the inverse of A obtaining the value of x as $x = A^{-1}b$. However, this calculation can be performed more efficiently by using the backslash operator or a specific matrix factorization. The example below shows the concept.

•• Example 14.

The following script creates a random system of linear equations that is solved in three different ways:

```
using LinearAlgebra

n = 1000
A = rand(n,n)
b = rand(n,1)

@time xa = inv(A)*b

@time xb = A\b

L,U,p = lu(A)
@time begin
    y = L\b[p]
    xc = U\y
end
```

Here, `@time` is a macro for measuring calculation time. First, a random matrix A and a random vector b are generated. Next, the system is solved by three methods: direct calculation of the inverse, backslash operator and the LU factorization. The first two methods are intuitive. The third method is based on the following factorization:

$$PA = LU \tag{37}$$

where L is a lower triangular matrix, and U an upper triangular matrix; P is a permutation matrix. Then, the linear system can be solved in two steps: first, the system $Ly = Pb$ is solved, by creating a new vector y for the intermediate calculation; next, the system $Ux = y$ is solved to obtain the vector x . Notice that, Julia returns a permutation vector instead of a permutation matrix, so the product Pb can be efficiently calculated as `b[p]`. The student is invited to execute this code and compare performance.

If the matrix is singular, we can still obtain a solution for the linear system (36) using the Moore-Penrose pseudoinverse. The pseudoinverse is represented as A^+ and calculated in Julia by the command `pinv()`. Singularity of A implies that the solution either does not exist or it is not unique. Thus, the pseudoinverse provides the least squares solution to the linear system, that is, a solution with a minimum Euclidean norm. In this case, the value of x is the best possible vector associated to the following optimization problem:

$$\min \|Ax - b\|^2 + \mu \|x\|^2 \quad (38)$$

where $\mu \rightarrow 0$. Mathematical properties of the pseudoinverse can be studied in [Barata and Hussein, 2012]

•• Example 15.

The script presented below is a simple calculation for a linear system with a singular matrix.

```
using LinearAlgebra
A = [1 3; 2 6]
b = [8;16]
x = pinv(A)*b
println(x)
println(A*x-b)
```

We can prove the matrix is singular by calculating its determinant.

On the other hand, most of the admittance matrices in power systems are highly sparse. This type of matrix has a characteristic structure where most of its entries are zero, as depicted in Figure 5. In this case, we can efficiently store and handle the matrix using the package `SparseArrays`. Calculations with a sparse matrix may be done efficiently since it has fewer actual data points compared to dense matrices. When performing calculations on sparse matrices, algorithms can take advantage of this sparsity by skipping operations involving zero elements. This reduces the overall number of computations needed, leading to faster computation times. Using sparse arrays is simple, as demonstrated in the example below.




Figure 5: Visualization of a sparse matrix. Dots represent entries different from zero.

•• Example 16.

The following script creates a sparse matrix $A = M + D$, where each entry in M is generated by rounding a random number between 0 and 0.6, and, D is a random diagonal matrix resulting in a sparse matrix. The rest of the code is intuitive.

```
using LinearAlgebra
using SparseArrays
n = 10
M = round.(0.6*rand(n,n))
D = Diagonal(rand(n))
A = sparse(M+D)
display(A)
```

6 Data frames

Power distribution networks may have thousands of nodes and lines, which imply large amount of data. Therefore, it is required to efficiently store and handle information through data frames. This data structure organizes the information into a 2-dimensional table of rows and columns, similarly to a spreadsheet. The package `DataFrames` allows intuitively handling this type of structure, as exemplified below.

●● Example 17.

The information of four substations is stored in three vectors that contain the identification name, the voltage level in kV and, whether it is connected to the system. This information is registered in vectors `A`, `B`, `C`. Then, a data frame is created. Finally, the information of the data frame is printed, first as the entire data frame and next individually.

```
using DataFrames
A = ["Electron platyrhynchum",
     "Guacamaya",
     "Motmot",
     "Trochilidae"]
B = [500;500;115;230]
C = [true, true, true, false]
df = DataFrame(Name=A,
               Voltage_level=B,
               Connected=C)

println(df)
# accesing individual data
println(df.Name[1])
println(df.Voltage_level[3])
using Plots
bar(df.Name, df.Voltage_level)
```

Notice the data-frame can store different type of data (e.g., text, numbers and boolean) in the same structure. The last line combines the package `DataFrame` with the package `Plots` to obtain a bar plot. The student is invited to execute the code above to see the way as a data frame organizes the information.

It is common in power systems applications to storage the information of the test systems using a comma separated values or a CSV file. This is a plain text file that stores data by delimiting data entries with commas. Julia is able to read said file, transforming into a data frame object. For that, we require the packages `CSV` and `DataFrames`. Let us consider the following practical example:

●● Example 18.

Let us consider a CSV file with the data presented in Table 1, which corresponds to the CIGRÉ benchmark for medium voltage distribution network [Rudion et al., 2006]. In this table, the values of `R` and `X` are in Ohm/km, `C` in nF/km and `Length` in km. We named this file as `CigreMicrogrid.csv`, which is loaded with the following script, saved in the same folder as the script:

```
using CSV
using DataFrames
microgrid = DataFrame(CSV.File("CigreMicrogrid.csv"))
print(microgrid)
```

Entries in a data frame can be obtained by the index number; for example, `microgrid[1,3]` returns 0.579 corresponding to the resistance of the first line section which connects nodes 1 and 3. The same value can be obtained as `microgrid.R[1]`. In fact, all resistances can be obtained as an array given by `microgrid.R[:]`. However, printing a dataframe results in neater output than printing an array. We finish this example by visualizing the values of `R` as a bar plot, as given below:

```
using Plots
bar(microgrid.R, label="Resistance",
    xlabel="Node", ylabel="Ohm/km")
```

Table 1: Data for the CIGRÉ medium voltage distribution networks taken from [Rudion et al., 2006]

From	To	R	X	C	Length
1	2	0.579	0.367	993	2.82
2	3	0.164	0.113	413	4.42
3	4	0.262	0.121	405	0.61
4	5	0.354	0.129	285	0.56
5	6	0.336	0.126	343	1.54
6	7	0.256	0.130	235	0.24
7	8	0.294	0.123	350	1.67
8	9	0.339	0.130	273	0.32
9	10	0.399	0.133	302	0.77
10	11	0.367	0.133	285	0.33
3	8	0.172	0.115	411	1.30
11	4	0.423	0.134	310	0.49

7 Good practices

Writing a general purpose software is different from writing a scientific software. However, there are some concepts from software engineering that may be useful to obtain a clean and readable script. Here, we present some recommendations inspired in [Wilson et al., 2017]:

Planning: think about the architecture of the script before writing the code. Decide how many functions are required and how the information will flow among these functions. Drawing on a piece of paper may help to see the entire architecture, thinking about the future and how the script may growth.

Writing: place a brief explanatory comment at the start of every program and function. Give functions and variables meaningful names. Julia allows including Greek letters and tildes as variables. This is a very useful feature that should be used to the fullest. Include a README file with details about the project. Markdown is an excellent tool for this purpose.

Organize the input data: It is important to properly organize CSV or Excel databases before running any code. Organize the data as simple as possible, similarly to a Dataframe. Choose meaningful names for each variable. Use a single word without spaces for each variable. For example, R_{pu} instead of R or R_{pu} to represent resistance when it is given in per unit. See for example [Broman and Woo, 2018] for recommendations about data organization.

Create a toy-model: do not pretend to solve a large test system without solving first a small “toy-model”. For instance, you may consider first a small DC distribution network with three or four nodes, or an AC system without reactive power, before considering the full problem.

Decompose programs into functions: Breaking code into functions allows creating modular components that may be used in further applications. For example, it is useful to create functions for calculating the Y_{bus} which is used in almost every power system application. These functions should be small and easy to read. The Julia compiler can often better optimize individual functions compared to a single large script.

8 Further readings

There is never enough linear algebra for a power systems engineer. Several books, web pages and even videos address to this subject. For example, the work of [Darve and Wootters, 2021] presents a complete course of linear algebra with applications in Julia. A more formal treatment can be found in books like [Axler et al., 2009], [Hubbard and Hubbard, 1999], and [Horn and Johnson, 1985]. It is also important to learn how to transform mathematical concepts into useful algorithms (this is one of the primary motivations of this book). However,

writing clean code is more an art than a science. There is plenty of material available in books and on the Internet. For instance, [Martin and Coplien, 2009] presents a comprehensive introduction to software engineering. Nevertheless, it is essential to remember that power systems' computation is a field with particular characteristics. What works in an accounting software may not be relevant in power systems applications. The experience helps to develop better scripts, so, it is recommended to try different approaches until finding the best practices.

References

- [Axler et al., 2009] Axler, S., Gehring, F., and Ribet, K. (2009). *Linear algebra done wright*. Springer, NY, 2 edition.
- [Azimzadeh and Forsyth, 2016] Azimzadeh, P. and Forsyth, P. A. (2016). Weakly chained matrices, policy iteration, and impulse control. *SIAM Journal on Numerical Analysis*, 54(3):1341–1364.
- [Barata and Hussein, 2012] Barata, J. and Hussein, M. (2012). The moore–penrose pseudoinverse: A tutorial review of the theory. *Brazilian Journal of Physics*, 42:146–165.
- [Broman and Woo, 2018] Broman, K. W. and Woo, K. H. (2018). Data organization in spreadsheets. *The American Statistician*, 72(1):2–10.
- [Darve and Wootters, 2021] Darve, E. and Wootters, M. (2021). *Numerical Linear Algebra with Julia*. Society for Industrial and Applied Mathematics, Philadelphia, PA.
- [Garces, 2022] Garces, A. (2022). *Convex optimization*, pages 39–59.
- [Horn and Johnson, 1985] Horn, R. A. and Johnson, C. R. (1985). *Matrix Analysis*. Cambridge University Press.
- [Hubbard and Hubbard, 1999] Hubbard, J. H. and Hubbard, B. B. (1999). *Vector Calculus, Linear Algebra, And Differential Forms A Unified Approach*. Prentice Hall.
- [Martin and Coplien, 2009] Martin, R. C. and Coplien, J. O. (2009). *Clean code: a handbook of agile software craftsmanship*. Prentice Hall, Upper Saddle River, NJ [etc.].
- [Rudion et al., 2006] Rudion, K., Orths, A., Styczynski, Z., and Strunz, K. (2006). Design of benchmark of medium voltage distribution network for investigation of dg integration. In *2006 IEEE Power Engineering Society General Meeting*, pages 6 pp.–.
- [Wilson et al., 2017] Wilson, G., Bryan, J., Cranston, K., Kitzes, J., Nederbragt, L., and Teal, T. K. (2017). Good enough practices in scientific computing. *PLOS Computational Biology*, 13(6):1–20.